

An Introduction to the ODS Destination for Word

David W. Kelley, SAS Institute Inc.

ABSTRACT

The SAS® Output Delivery System (ODS) destination for Word enables customers to deliver SAS® reports as native Microsoft Word documents. The ODS WORD statement generates reports in the Office Open XML Document (.docx) format, which has been standard in Microsoft Word since 2007. The .docx format uses ZIP compression, which makes for a smaller storage footprint and speedier downloading. ODS WORD is preproduction in the sixth maintenance release of SAS 9.4.

This paper shows you how to make a SAS report with ODS WORD. You learn how to create your report's content (images, tables, and text). You also learn how to customize aspects of your report's presentation (theme and styles). And you learn how to enhance your report with reader-friendly features such as a table of contents and custom page numbering. If you're cutting and pasting your SAS output into Microsoft Word documents, then this paper is especially for you!

INTRODUCTION

For almost as long as ODS has existed, customers have used it to automatically generate SAS reports as Microsoft Office (Excel, PowerPoint, Word) documents. Prior to Office 2007, Microsoft used proprietary binary formats for these documents. ODS developers worked around the restricted formats by inventing destinations (ODS RTF, ODS TAGSETS.RTF, ODS TAGSETS.EXCELXP, ODS TAGSETS.MSOFFICE2K) that generate reports in intermediate text formats (RTF, Excel XML, HTML) ingestible by Word and Excel.

This indirect approach has drawbacks. Customers are required, within an Office application, to manually save the generated document in the standard format. The intermediate formats are not as rich as the standard formats, so SAS reporting elements might get lost in translation. For example, Excel XML does not support images, so SAS/GRAPH and ODS GRAPHICS do not work with ODS TAGSETS.EXCELXP. And third-party applications do not support the intermediate formats well, if at all. Have you ever tried to view an RTF document on an iPad or iPhone? Good luck with that!

Microsoft released the Office Open XML file formats in 2007, and they were standardized in 2008. We in ODS R&D knew then that we wanted to build a new suite of destinations for the new Office. We also knew that we wanted to start with PowerPoint, because we had no existing destination for PowerPoint. We plugged that hole in our lineup with ODS POWERPOINT, which came out in SAS 9.4. Three maintenance releases later, we followed up with ODS EXCEL, which has been hugely popular with customers ever since.

With SAS 9.4M6, we completed our Office portfolio by adding ODS WORD. While similar to ODS RTF in its user interface (UI), ODS WORD offers new capabilities by virtue of its support of the native Word .docx format. Later we'll explore several of these capabilities, including themes and page numbering. But first we need to learn how to use ODS WORD in a simple SAS program. Crawl before you walk!

MUSIC TO MY EARS

This paper focuses on output, specifically ODS WORD output. But we start with input, specifically data for our ODS WORD reports. I wanted data that is derived from real life experience, is unique, and hopefully is interesting.

I chose my iTunes music playlist as the input data. A playlist is singular and empirical by nature. It has the potential to be interesting, if for no other reason than to provoke the argument that there is no accounting for taste in music!

In my PC iTunes app, I synchronized my iPod and exported the playlist as a text file. Then, using the IMPORT procedure and DATA step, I imported the text file into a SAS data set called WORK.MYITUNES.

Table 1 displays the variables for WORK.MYITUNES.

Variable	Type	Format
Album	Char	\$95.
Artist	Char	\$255.
Composer	Char	\$118.
Date_Added	Num	DATETIME.
Genre	Char	\$18.
Last_Played	Num	DATETIME.
Last_Skipped	Num	DATETIME.
Name	Char	\$456.
Plays	Num	BEST12.
Skips	Num	BEST12.
Time	Num	MMSS.
Track_Count	Num	BEST12.
Track_Number	Num	BEST12.
Year	Char	\$4.

Table 1. WORK.MYITUNES Variables

WORK.ITUNES is a collection of tracks. I know that the tracks are comprised mostly of songs, although there is a scattering of instrumental and spoken word recordings among them. Note that one song might be represented by more than one track. (It could appear on multiple albums and/or have multiple versions.)

EASY AS PIE

What types of music are in my playlist? We can answer that question broadly by summarizing the tracks by genre. Here is one way to do that with ODS WORD:

```
ods word file="c:\users\sasdck\onedrive - sas\genres.docx";
title "My iTunes Genres";
proc sgpie data=myitunes;
  pie genre / datalabeldisplay=(percent);
run;
ods word close;
```


Like other ODS destinations, the ODS WORD destination supports the "ODS sandwich" programming idiom. You sandwich your procedure and DATA steps in ODS WORD statements. The beginning ODS WORD statement opens an ODS WORD destination and prepares it to process the procedure and DATA step output. The ending ODS WORD statement closes the destination, which terminates the generation of output to the specified file. An open ODS WORD destination remains open until it is explicitly closed or the SAS session ends.

The FILE= option specifies the output file name. If you do not specify a name, ODS WORD supplies a default name (typically sasword.docx). You should always give the file an extension of .docx because that is what Microsoft Windows and Word expect.

The code writes the output file to the [Microsoft OneDrive](#) directory on my PC. Windows automatically synchronizes files written to that location to my OneDrive cloud storage. This is handy for accessing the files on mobile devices. We will examine that capability in the [GO MOBILE](#) section.

The SGPIE procedure generates pie and donut charts. PROC SGPIE is preproduction in SAS 9.4M6. Display 1 below shows the pie chart representation of my playlist genres in desktop Microsoft Word.

My music profile is representative of the Baby Boomer generation (1946 to 1964), of which I am a member, barely. Just recently the Millennial generation surpassed the Boomers as the largest generation in the US population. A **Millennial's** pie chart would look quite different from mine, don't you think?



Display 1. iTunes Playlist Genres (Pie Chart)

STACKS OF TRACKS

The pie chart is a fine starting point. It invites follow-up questions. How many tracks are there? And what genres comprise the Other category, which is the fourth largest? We can answer those questions with the TABULATE procedure:

```
ods word file="c:\users\sasdck\onedrive - sas\genres2.docx";
title "My iTunes Genres";
proc tabulate data=myitunes;
  class genre;
  table genre=" " all="Total", n="Tracks" colpctn="%" /box="Genre";
run;
ods word close;
```

Display 2 shows the crosstabular representation of my playlist genres. We see that tracks classified as Folk, Jazz, Religious, and Traditional represent the lion's share of the pie chart's Other category. This is $45+65+25+31=166$ of 2,659 tracks, or 6.2%.

The screenshot shows a Microsoft Word document titled 'genres2.docx' with the following table:

Genre	Tracks	%
Alternative	4	0.15
Alternative Rock	2	0.08
Blues	20	0.75
Children's	18	0.68
Classical	10	0.38
Country	27	1.02
Country & Folk	536	20.16
Folk	45	1.69
Folk-Rock	108	4.06
Jazz	65	2.44
Miscellaneous	1	0.04
Pop	229	8.61
R&B	458	17.22
Religious	25	0.94
Rock	895	33.66
Soundtrack	185	6.96
Traditional	31	1.17
Total	2659	100.00

Display 2. iTunes Playlist Genres (Crosstabular Table)

The other catch-all genre we will explore further is Soundtrack. Saying that a track is from a soundtrack tells you nothing about what type of music, if any, it represents. And there isn't a WORK.MYITUNES variable for subgenre. But we can list the soundtrack albums, and that will tell you all you really need to know. (Well, that and perhaps Wikipedia 😊.)

The playlist is structured by track, not album. This means that we must filter the playlist before we can list the albums. We filter with the SORT procedure, and then we list with the ODSLIST procedure:

```
proc sort data=myitunes(keep=album genre where=(genre="Soundtrack"))
    out=soundtracks nodupkey;
    by album;
run;


ods word file="c:\users\sasdck\onedrive - sas\soundtracks.docx";
title "My iTunes Soundtracks";
proc odslist data=soundtracks;
    item put(album,$95.)/style={liststyletype=decimal};
run;
ods word close;
```

The PROC ODSLIST ITEM statement accepts an expression, which is evaluated for each observation in the input data set. We know from [Table 1](#) to specify a format of \$95. to the PUT function for variable Album.

By default, lists are unordered (bulleted). I want an ordered list, so I specify LISTSTYLETYPE=DECIMAL as a style attribute override.

What are the results? Display 3 tells the tale—the listed soundtracks are all musicals. Some are Broadway cast recordings, some are movie soundtracks, and others are not discernible from the album title.

The report lists eleven soundtrack albums, but arguably it should be thirteen. My playlist includes the soundtrack to the movie *Purple Rain*, but iTunes assigns its genre to be R&B instead of Soundtrack. And the soundtrack to the movie *O Brother, Where Art Thou?* has no assigned genre at all. Your output is only as good as your input data!



Display 3. iTunes Playlist Soundtracks

NOW PLAYING

So far, we have explored the types of music in my playlist. But what tracks have been played, and how many times? More specifically, which artists, by genre, have gotten the most plays?

A Top N report is tailor-made to answer these types of questions. If you are not familiar with Top N reports, then refer to the SAS Technical Support sample ([Creating a Top N Report](#)) on the subject. My Top N report is adapted from the code in the sample. The main steps of the algorithm are as follows:

1. Summarize the WORK.MYITUNES number of plays by genre and artist, and then store the results in output data set WORK.SUMMARY.
2. Sort WORK.SUMMARY so that the top plays by genre appear at the top of output data set WORK.TOPN.
3. Rank the WORK.TOPN top 10 plays by genre.
4. Print the WORK.TOPN top 10 artists in plays by genre.

In addition, we instruct ODS WORD to automatically generate a clickable table of contents (TOC) for the report.

Here is the code for the Top N report:

```
proc means data=myitunes sum noprint;
  var plays;
  class genre artist;
  output out=summary sum=plays genre /levels;
run;

proc sort data=summary out=topn;
  where _type_>2;
  by genre descending plays;
run;

data topn;
  length rank 8;
  label rank="Rank";
  set topn;
  by genre descending plays;
  if nmiss(of plays) then delete;
  if first.genre then rank=0;
  rank+1;
  if rank le 10 then output;
run;

ods word file="c:\users\sasdck\onedrive - sas\topn.docx"
  options(contents="on" toc_data="on" keep_next="on");
title "Top 10 Artists in Plays by Genre";
ods proclabel=" ";
proc print data=topn noobs contents="" label;
  by genre;
  var rank artist plays;
run;
ods word close;
```

ODS WORD accepts certain user options via the OPTIONS suboption. To enable TOC creation, we specify CONTENTS="ON" and TOC_DATA="ON". KEEP_NEXT="ON" tells ODS WORD to try to keep each BY line on the same page as the table that immediately follows it in the report. PROC PRINT places multiple BY groups on a page, and we do not want a BY line and its associated table to be separated by a page break. Such lack of fit-and-finish in a report is a dead giveaway that it is computer-generated. We can do better than that!

The ODS PROCLABEL statement instructs ODS to omit the procedure label ("The Print Procedure") from the TOC. The label adds no useful information and is just visual clutter. Similarly, the PROC PRINT CONTENTS= setting omits the output object label ("Data Set WORK.TOPN") from the TOC. Less is more.

Display 4 below shows the Top N report. The TOC is rendered at the beginning of the report and is followed by the Top 10 rankings, which are grouped alphabetically by genre. Some genres have fewer than 10 artists, which suggests that I should also make a Top N report across genres. In case you are wondering, the artist with most plays across genres is Ray Charles (865), followed by The Beatles (531), and Elton John (509).

Note that Microsoft Word does not automatically update the TOC when the document is loaded. You must instruct Microsoft Word to update the TOC manually. One way to do that is to type CTRL+A F9 on your keyboard.

Table of Contents

Genre=Blues 1

Genre=Children's 1

Genre=Classical 1

Genre=Country 1

Genre=Country & Folk 2

Genre=Folk 2

Genre=Folk-Rock 2

Genre=Jazz 2

Genre=Pop 3

Genre=R&B 3

Genre=Religious 3

Genre=Rock 4

Genre=Soundtrack 4

Genre=Traditional 4

11:24 Thursday, February 21, 2019 1

Top 10 Artists in Plays by Genre

Genre=Blues

Rank	Artist	plays
1	B.B. King	60
2	B.B. King & Bobby Bland	8
3	Little Walter	3
4	Buddy Guy	2
5	Etta James	2
6	Howlin' Wolf	2
7	Jimmy Rogers	2
8	John Lee Hooker	2
9	Koko Taylor	2
10	Muddy Waters	2

Genre=Children's

Rank	Artist	plays
1	Disney	24
2	Julie Andrews	8
3	David Tomlinson/Dick Van Dyke/The Londoners	7
4	Amy Lou Barnes, Sally Mueller, Mary Moder, Alice Sizer, Betty Bayne	6
5	Burl Ives	2
6	Dickie Jones	2
7	Peggy Lee	2
8	Peggy Lee & The Mello Men	2
9	Adriana Caselotti	1
10	Cliff Edwards, Jim Carmichael & The Hall Johnson Choir	1

Genre=Classical

Rank	Artist	plays
1	All-Time Greatest Hits	165

Genre=Country

Rank	Artist	plays
1	Johnny Cash	112

Display 4. iTunes Top 10 Artists in Plays by Genre

In the next section, we shift our focus from the content of an ODS WORD report to its presentation. We want our reports to be print-ready right out of SAS—no manual post-processing or editing should be required. A little extra time and effort invested in SAS code can save repeated interventions by hand downstream.

A MATTER OF STYLE

Like most ODS destinations, ODS WORD accepts a named *style* that determines presentation aspects of reports (fonts, colors, and so on). The content of the report primarily comes from the executed procedure and DATA steps, as well as from ODS and global statements. The look and feel of the report comes primarily from the style. If you want to customize the look and feel, then you need to modify the style.

The default ODS WORD style is named Styles.Word. Style names are case-insensitive.

A style is organized as a collection of named *style elements*. Each style element comprises a collection of name-value pairs called *style attributes*. You can think of a style as a bundle of settings, where logically related settings are grouped into elements and defined as attributes. Style element names are case-insensitive.

A style can inherit elements and attributes from a parent style. Style inheritance enables convenient modification of an existing style. A complete style has many elements and attributes, and you will get your custom style functioning much faster if you begin with an existing style, rather than trying to make a style from scratch. That way you can focus on just the attributes that you need to change.

Knowing which elements and attributes to change can be challenging. The ODS style documentation is indispensable to the endeavor. Plus, SAS Technical Support is always ready and able to help you!

But you do not need to worry about that now. My ODS WORD style examples show you which elements and attributes to modify, and what effect modification has on the report look and feel. The first example highlights a common user reporting scenario: customizing page numbers.

PAGE X OF Y

ODS WORD generates page numbers in the Word document if the NUMBER system option is in effect. More precisely, ODS WORD generates an instruction that tells Microsoft Word to compute the page numbers when the document is loaded. Microsoft Word paginates the document at that time, and that is when the page numbers are determined.

By default, ODS WORD right-justifies the page number in the document header, which is repeated at the top of each page. The starting value of the page number is controlled by the PAGENO system option, which defaults to 1. [Display 1](#) shows the default ODS WORD page numbering behavior.

Users often wish to customize the position and/or text of the page number. Changing the position typically involves left-justifying or center-justifying the page number in the header, or placing the page number in the document footer instead of the header. ODS WORD supports six positions: left, center, and right in the header and footer.

A common reason to modify the page number text is to define Page X of Y behavior. For a 3-page Word document, the desired text would be "Page 1 of 3", "Page 2 of 3", and "Page 3 of 3". Also popular is X of Y, which would be "1 of 3", "2 of 3", and "3 of 3". Ideally the customer could specify the text. Can ODS WORD do that, and if so, how?

Yes, it can! But first you must modify the PageNo style element, which governs page numbers in ODS WORD. And to do that, you need to use the TEMPLATE procedure, which is for creating and modifying ODS styles. The SAS code is below:

```
ods path (prepend) work.templat(update);
proc template;
define style styles.mystyle;
  parent = styles.word;
  class pageno /
    content = "Page {PAGE} of {NUMPAGES}"
    just = c
    vjust = b
  ;
end;
run;
```

PROC TEMPLATE compiles style definitions into binary form and stores them in a SAS library member called an *itemstore*. The ODS PATH statement controls which itemstore receives the compiled template. In this example, the Styles.MyStyle definition is compiled into WORK.TEMPLAT, which means that it persists no longer than the SAS session that creates it.

For the reason discussed previously, Styles.MyStyle is defined to inherit from Styles.Word. The PROC TEMPLATE PARENT= option makes this happen. All that is left to do is to update the PageNo style element to specify the text and position of page numbers. You do that with the CLASS statement. CLASS accepts a style element name and one or more style attributes. The specified style attribute values override any that are inherited from the parent style.

In the example, the CONTENT attribute is set to "Page {PAGE} of {NUMPAGES}". {PAGE} and {NUMPAGES} are [field codes](#) that Microsoft Word automatically updates when it paginates the document. The CONTENT setting overrides the default value of "{PAGE}". The rest of the text is static.

The JUST attribute governs horizontal justification of the page number text. Likewise, VJUST governs vertical justification. The code requests that page numbers be centered in the footer (bottom) of the Word document.


Now that Styles.MyStyle is defined, we can use it to customize a report. How about we update our Top N report from before? The updated code follows:

```
ods word file="c:\users\sasdck\onedrive - sas\topn2.docx"
  options(keep_next="on") style=mystyle;
title "Top 10 Artists in Plays by Genre";
ods proclabel=" ";
proc print data=topn noobs contents="" label;
  by genre;
  var rank artist plays;
run;
ods word close;
```

The ODS WORD STYLE= option is how we tell ODS WORD to use Styles.MyStyle instead of defaulting to Styles.Word. Note that you do not need to specify the "Styles." prefix; it is syntactic sugar.

The only other code difference from before is that we do not request a TOC. That is because the example is about page numbers, and the TOC does not have a page number.

Display 5 shows the updated Top N report, which is zoomed to 150% in Microsoft Word.



Display 5. Page X of Y Page Numbering

Dates work similarly to page numbers in ODS WORD. Space does not permit a full treatment of dates in this paper. I will scatter these breadcrumbs for adventurous users wishing to experiment with custom dates:

- By default, the style element for dates is BodyDate. Microsoft Word supplies the date value dynamically when the document is loaded.
- By default, the value of the CONTENT attribute is '{ DATE \@ "HH: mm dddd, MMMM dd, yyyy"}'. You will likely want to modify that value .
- If you use the ODS WORD SASDATE option, ODS WORD supplies the date value, which is static. The associated style element is SASDate. In this case, the CONTENT attribute is ignored.

Can you see the possibilities for customizing page numbers and dates in your ODS WORD reports? There is scarcely a limit to what you can do, once you familiarize yourself with field codes. Happy hunting!

THEMES AND SCHEMES

Microsoft Office applications have *themes* for controlling visual aspects of both the application UI and the loaded document. Document themes control colors and fonts in the document. In the UI, you can easily change the appearance of your document by changing the associated theme. The scope of effect is global, which is handy for quickly assigning a uniform look and feel to your documents.

Documents created with ODS WORD automatically work with Microsoft Word themes. That is because the Styles.Word style includes a theme definition. The Styles.Word theme is designed to mimic the default theme in Microsoft Word 2016/365. Remember, you win if your reader cannot tell that your ODS WORD report is computer-generated ☺.

To browse themes in Microsoft Word, select Themes on the Design tab. Display 6 below shows the playlist report from [Display 2](#), but using the Banded theme instead of the ODS WORD default theme.

The screenshot shows the Microsoft Word interface with the 'Design' tab active. The 'Themes' task pane is open on the left, displaying various theme options. The 'Banded' theme is selected. The main document area shows a table titled 'My iTunes Genres' with the following data:

Genre	Tracks	%
Alternative	4	0.15
Alternative Rock	2	0.08
Blues	20	0.75
Children's	18	0.68
Classical	10	0.38
Country	27	1.02
Country & Folk	536	20.16
Folk	45	1.69
Folk-Rock	108	4.06
Jazz	65	2.44
Miscellaneous	1	0.04
Pop	229	8.61
R&B	458	17.22
Religious	25	0.94
Rock	895	33.66
Soundtrack	185	6.96
Traditional	31	1.17
Total	2659	100.00

At the bottom of the Word window, the status bar shows: Page 1 of 1, 60 words, English (United States).

Display 6. iTunes Playlist Genres (Banded Theme)

You can view the Styles.Word theme definition by running PROC TEMPLATE, as shown in Output 1.

```
? proc template; source styles.word; run;

define style Styles.Word;
  scheme 'ODS Word' /
    followed_hyperlink = cx954f72
    hyperlink = cx0563c1
    accent6 = cx70AD47
    accent5 = cx5B9BD5
    accent4 = cxFFC000
    accent3 = cxA5A5A5
    accent2 = cxED7D31
    accent1 = cx4472C4
    light2 = cxe7e6e6
    dark2 = cx44546A
    light1 = cxFFFFFF
    dark1 = cx000000
    body_font = ("Calibri, <sans-serif>, <MTsans-serif>",11pt)
    heading_font = ("Calibri Light, Calibri, <sans-serif>, <MTsans-
serif>",16pt,bold);

[... rest of style definition omitted ...]

end;
NOTE: Path 'Styles.Word' is in: SASHELP.TMPLMST.

NOTE: PROCEDURE TEMPLATE used (Total process time):
      real time          0.42 seconds
      cpu time           0.15 seconds
```

Output 1. Styles.Word Theme Output

Themes provide these color definitions:

- Two light text/background colors
- Two dark text/background colors
- Six accent colors: for GraphData1 - GraphData6 in ODS statistical graphs (accent1 is also used for headings)
- Two hyperlink colors: for unfollowed and followed hyperlinks

You use the PROC TEMPLATE SCHEME statement to define theme colors. You can readily visualize the colors with Microsoft Word. In Word 2016/365, select the Design tab, Colors, and then Customize Colors.... If you have an ODS WORD document loaded, then the Styles.Word colors appear as shown in [Display 7](#).

The Microsoft Word theme colors UI includes a color picker, which can provide the RGB decimal values for the selected color. The SCHEME statement expects hexadecimal values, so you will have to convert the RGB values from the color picker. [Display 8](#) shows the color picker with the Styles.Word accent1 color selected.

Themes support two types of fonts:


- Heading font: for headings, system titles and footnotes, BY lines, and procedure titles
- Body font: for table cells, paragraphs, and lists

The SCHEME statement allows you to specify heading and body fonts. For Styles.Word, they are Calibri Light and Calibri respectively, if available. Calibri is the standard font family for Microsoft Office applications.


Calibri and Calibri Light are Windows fonts. They are not available to SAS running on UNIX or OS/390 platforms. On those systems, ODS WORD substitutes a font that is supplied by SAS, likely Albany AMT, for Calibri and Calibri Light.

If you want to customize your ODS WORD theme, then you need to create a style that inherits from Styles.Word, and update the SCHEME statement with your desired colors and/or fonts. In general, when creating a style for use with ODS WORD, the style should always inherit from Styles.Word. That way it is sure to have a theme definition. ODS WORD expects a theme definition, and if it does not get one, it will likely generate unexpected results. Fair warning!

If you find that you regularly substitute a particular Microsoft Word theme for the default ODS WORD theme, then you can create a custom ODS WORD style that mimics your favored Microsoft Word theme. Sorry, ODS WORD will not do this for you automatically! Display 7 and Display 8 show the ODS Word theme colors and Microsoft Word color picker.



Display 7. ODS WORD Theme Colors



Display 8: Microsoft Word Color Picker

In the next section, we demonstrate ODS WORD reports on mobile devices. Escape the desktop!

GO MOBILE


If you use Microsoft Word 2016/365, then you have access to [Microsoft OneDrive](#), which is the Microsoft cloud file storage and synchronization service. OneDrive, like similar products from other vendors (Apple iCloud, Google Drive, Dropbox), makes it easy to share files across PCs and mobile devices. My go-to device for accessing work-related documents away from the office is an iPad. Naturally I wanted to be able to view ODS WORD documents on my iPad. But before I could do that, I had to install free client software on my PC and iPad:

- [OneDrive sync client for Windows](#)
- [OneDrive app for iOS](#)
- [Microsoft Word app for iOS](#)

When the OneDrive sync client was installed on my PC, it created the Windows directory C:\Users\sasdck\OneDrive - SAS. Any files or folders written to that location are synchronized to my OneDrive cloud storage. That makes them available to the OneDrive and Word apps on my iPad.



The examples for this paper use this technique; the SAS code instructs ODS WORD to write the .docx files to my OneDrive directory. Up to this point I have shown how Microsoft Word on my PC renders the ODS WORD documents. How about with the Word app on my iPad?

Display 9 shows the Word app in action. When you launch it, it lists the places it knows to look for Word documents. I have selected *OneDrive - SAS*, which lists the synced folders and files, including the Word documents for the examples.



Display 9. iPad Word App Documents View

When I select the *genres* Word document, the app renders its contents, as seen in Display 10. Compare to the Windows version shown in [Display 1](#).



Display 10. iPad Word App Genres (Pie Chart)

That is PROC SGPIE output from my PC to my iPad via the cloud. It is pie in the sky, figuratively speaking!

CONCLUSION

The ODS WORD destination is preproduction software in SAS 9.4M6. Development is well underway on the production software version of ODS WORD. That version will introduce two advanced new features:

- Multi-column layout (ODS WORD COLUMNS=)
- Gridded layout (ODS LAYOUT GRIDDED)

There will be other new features as well.

There is more to learn about ODS WORD. This paper has demonstrated how to get started. The next step is up to you. Try making some ODS WORD reports. We expect that you will be pleased with the results. Let us know how it goes!

REFERENCES

- Hemedinger, Chris. "Sample 33009: Creating a Top N Report." Cary, NC: SAS Institute Inc. August 26, 2008. Available at <https://support.sas.com/kb/33/009.html>.
- "List of field codes in Word." Microsoft. Available at <https://support.office.com/en-us/article/list-of-field-codes-in-word-1ad6d91a-55a7-4a8d-b535-cf7888659a51>.

ACKNOWLEDGMENTS

The author would like to thank Allison Crutchfield, Jane Eslinger, Kathryn McLawhorn, Chevell Parker, and Sara Thomas of SAS for reviewing this paper.

RECOMMENDED READING

Hunter, Tim. "A First Look at the ODS Destination for PowerPoint." *Proceedings of the SAS Global Forum 2013 Conference*. Cary, NC: SAS Institute Inc. Available at <https://support.sas.com/resources/papers/proceedings13/041-2013.pdf>.

CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the author at:

David W. Kelley
SAS Institute
100 SAS Campus Drive
Cary, NC 27513-2414
David.Kelley@sas.com

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

Other brand and product names are trademarks of their respective companies.